

Hewlett-Packard

HP-19C/HP-29C SOLUTIONS

CIVIL ENGINEERING

INTRODUCTION

This HP-19C/HP-29C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become an expert on your HP calculator.

You will find general information on how to key in and run programs under "A Word about Program Usage" in the Applications book you received with your calculator.

We hope that this Solutions book will be a valuable tool in your work and would appreciate your comments about it.

TABLE OF CONTENTS

1.

PROPERTIES OF CIRCULAR SECTIONS

1.

This program performs an interchangeable solution for four properties of circular sections. Given either the moment of inertia I, diameter d, polar moment of inertia J, or area A, the remaining properties can be calculated.



EQUATIONS:

$$I = \frac{\pi d^4}{64}$$

$$J = \frac{\pi d^4}{32}$$

$$A = \frac{\pi d^2}{4}$$

EXAMPLE 1:

If the moment of inertia of a section must be 60 in., what is the necessary diameter? What is the polar moment of inertia? What is the area?

EXAMPLE 2:

The diameter of a section is 10 centimeters. What is the moment of inertia? What is the polar moment of inertia? What is the area?

SOLUTIONS:

1.

```
  60.00 GSB1
  60.00  ***   (in.^4) I
          R↓
 120.00  ***   (in.^4) J
          R↓
  27.46  ***   (in.^2) A
          R↓
   5.91  ***   (in.)   d
```

2.

```
  10.00 GSB4
 490.87  ***   (cm^4)  I
          R↓
 981.75  ***   (cm^4)  J
          R↓
  78.54  ***   (cm^2)  A
          R↓
  10.00  ***   (cm)    d
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Input one of the following:				
	Moment of inertia	I	GSB	1	I
	Polar moment of inertia	J	GSB	2	I
	Area of section	A	GSB	3	I
	Diameter of section	d	GSB	4	I
3.	Roll the stack to see the results:				
	(Moment of inertia already on display)				
	Polar moment of inertia		R↓		J
	Area of section		R↓		A
	Diameter of section		R↓		d
4.	For a new case, go to step 2				

Program Listings

Step	Key	Comments
01	*LBL1	
02	Pi	$I = J/2$
03	÷	
04	√X	
05	2	
06	x	
07	GTO9	
08	*LBL2	J
09	2	
10	÷	
11	GTO1	
12	*LBL3	
13	Pi	A
14	÷	
15	GTO9	
16	*LBL4	
17	X²	
18	4	
19	÷	
20	*LBL9	$d^2/4$
21	STO1	
22	STO2	
23	ST×1	
24	4	
25	ST÷1	
26	x	
27	√X	d
28	RCL2	
29	Pi	
30	ST×1	
31	x	A
32	RCL1	
33	2	
34	x	J
35	RCL1	I
36	R/S	*** I J A d

*** "Print Stack" may be inserted before "R/S".

REGISTERS

0	1 Used	2 $d^2/4$	3	4	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

PROPERTIES OF RECTANGULAR SECTIONS

2.

This program performs an interchangeable solution for the moment of inertia I, the width b and the height h of a rectangular section. When b and h are known, the polar moment of inertia J and the section area can also be found.



EQUATIONS:

$$I = \frac{bh^3}{12}$$

$$J = \frac{bh(b^2+h^2)}{12}$$

$$A = b\,h$$

REMARKS:

Values of polar moment of inertia J calculated by this program must not be used to calculate torsional stress and strain in rectangular members.

EXAMPLE:

What is the moment of inertia of a section with b=3 and h=5? What is the polar moment of inertia? What is the area? What would b have to be if I=40?

SOLUTION:

```
 5.00 ENT↑
 3.00 ENT↑
 0.00 GSB1
31.25  ***   (in.⁴)
       R/S
15.00  ***   (in.²)
       R/S
42.50  ***   (in.⁴)
 5.00 ENT↑
 0.00 ENT↑
40.00 GSB1
 3.84  ***   (in.)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter the following (enter a zero for the				
	unknown quantity):				
	Height of section	h	ENT↑		
	Width of section	b	ENT↑		
	Moment of inertia of section	I			
3.	Compute the unknown quantity		GSB	1	I,b,or h
4.	(Optional) Compute the area of the section		R/S		A
5.	(Optional) Compute the polar moment of		R/S		J
	inertia				

Program Listings

Step	Key	Comment
01	*LBL1	
02	1	
03	2	
04	STO4	
05	R↓	
06	STO1	I
07	R↓	
08	STO2	b
09	X⇄Y	
10	STO3	h
11	X=0?	
12	GTO9	Calculate h
13	R↓	
14	X=0?	
15	GTO8	
16	RCL2	Calculate b
17	RCL3	Calculate I
18	3	
19	Y^X	
20	×	
21	RCL4	
22	÷	
23	GTO0	I
24	*LBL9	
25	RCL1	
26	RCL4	
27	×	
28	RCL2	
29	÷	
30	3	
31	1/X	
32	Y^X	
33	STO3	h
34	GTO0	
35	*LBL8	
36	RCL1	
37	RCL4	
38	×	
39	RCL3	
40	3	
41	Y^X	
42	÷	
43	STO2	b
44	*LBL0	** I,h, or b
45	R/S	
46	RCL2	
47	RCL3	
48	→P	
49	X^2	
50	RCL2	
51	RCL3	
52	×	
53	R/S	** A
54	×	
55	RCL4	
56	÷	
57	R/S	*** J

** "Printx" may replace "R/S".

*** "Printx" may be inserted before "R/S".

REGISTERS

0	1 I	2 b	3 h	4 12	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

PROPERTIES OF ANNULAR SECTIONS

This program provides an interchangeable solution for the moment of inertia I, the outside diameter d_o, and the inside diameter d_i of an annular section. Once d_o and d_i are known, the polar moment of inertia J and the area of the section can be calculated.



EQUATIONS:

$$I = \frac{\pi(d_o{}^4 - d_i{}^4)}{64}$$

$$J = \frac{\pi(d_o{}^4 - d_i{}^4)}{32} = 2I$$

$$A = \frac{\pi(d_o{}^2 - d_i{}^2)}{4}$$

EXAMPLE:

If d_i equals 3 inches and I equals 10 in^4, what is d_o? What is A?

What would I be if d_o equals 4.5 inches?

SOLUTION:

```
 3.00 ENT↑
 0.00 ENT↑
10.00 GSB1
 4.11  ***     do (in.)
        R↓
 3.00  ***     di (in.)
        R↓
10.00  ***     I (in.4)
 2.00   x
20.00  ***     J (in.4)
        R↓
 6.18  ***     A (in.2)

 3.00 ENT↑
 4.50 ENT↑
 0.00 GSB1
 4.50  ***
        R↓
 3.00  ***
        R↓
16.15  ***     I (in.4)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter the following (enter zero for the				
	unknown quantity):				
	Inside diameter	d_i	ENT↑		
	Outside diameter	d_o	ENT↑		
	Moment of inertia	I	GSB	1	d_o
3.	Roll the stack to see the results:				
	(Outside diameter already in display)				d_o
	Inside diameter		R↓		d_i
	Moment of inertia		R↓		I
	Polar moment of inertia		2	x	J
	Area of section		R↓		A
4.	For a new case, go to step 2				

Program Listings

Step	Key	Comment
01	*LBL1	
02	STO3	I
03	R↓	
04	STO1	d_o
05	X⇄Y	
06	STO2	d_i
07	X=0?	
08	GTO9	Calculate d_i
09	R↓	
10	X=0?	
11	GTO8	Calculate d_o
12	RCL1	Calculate I
13	4	
14	Y^X	
15	RCL2	
16	4	
17	Y^X	
18	-	
19	Pi	
20	x	
21	6	
22	4	
23	÷	
24	STO3	
25	*LBL7	
26	RCL1	
27	X²	
28	RCL2	
29	X²	
30	-	
31	Pi	
32	x	
33	4	
34	÷	A
35	RCL3	
36	RCL2	
37	RCL1	
38	R/S	*** d_o d_i I A
39	*LBL9	
40	RCL3	
41	CHS	
42	GSB0	
43	STO2	
44	GTO7	
45	*LBL8	
46	RCL2	
47	STO1	
48	RCL3	
49	GSB0	
50	STO1	
51	GTO7	
52	*LBL0	
53	6	
54	4	
55	x	
56	Pi	
57	÷	
58	RCL1	
59	4	
60	Y^X	
61	+	
62	4	
63	1/X	
64	Y^X	
65	RTN	

*** "Print Stack" may be inserted before "R/S".

REGISTERS

0	1 d_o	2 d_i	3 I	4	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

THIN-WALLED PRESSURE VESSELS

This program can be used to correlate diameter, stress, pressure and thickness for cylindrical and spherical pressure vessels. Either the hoop stress s_C or the longitudinal stress s_L may be input for cylinders. For spheres, only the hoop stress s_{sphere} is applicable.

Cut Cylindrical Section



Spherical Section



EQUATIONS:

for hoop stress in cylinders: $s_C = \frac{Pr}{t}$

for longitudinal stress in cylinders:

$$s_L = \frac{Pr}{2t}$$

for hoop stress in spheres: $s_{sphere} = \frac{Pr}{2t}$

where:

P is internal pressure;

D is diameter of vessel (r=D/2);

t is thickness of vessel

REMARKS:

The thickness of the walls must be negligible with respect to the value of the radius. The equations are not valid in the neighborhood of end closures for cylindrical vessels.

EXAMPLE 1:

A basketball has a diameter of 9.3 inches. The thickness of the cord layer which resists virtually all of the internal pressure is 1/32 inch. The recommended pressure is 9 pounds per square inch. What is the stress in the cord layer?

EXAMPLE 2:

A four inch diameter pipe contains steam at 1000 pounds per square inch. What thickness is required if hoop stress is not to exceed 15000 pounds per square inch?

SOLUTIONS:

1.
```
    9.30 ENT↑
    0.00 ENT↑
    9.00 ENT↑
   32.00  1/X
          GSB1
  669.60  ***    (psi)
```

2.
```
    4.00 ENT↑
15000.00 ENT↑
    2.00  ÷      sC/2
 1000.00 ENT↑
    0.00 GSB1
    0.13  ***    (in)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter the following (enter zero for the				
	unknown quantity):				
	Vessel diameter	D	ENT↑		
	Hoop stress/2 or longitudinal stress for	$\frac{s_c}{2}$, s_L, s_s	ENT↑		
	a cylinder or hoop stress for a sphere				
	Pressure	P	ENT↑		
	Wall thickness	t			
3.	Calculate the unknown quantity		GSB	1	D, $\frac{s_c}{2}$, s_L,
					s_s, P, or t
4.	For a new case, go to step 2				

Program Listings

```
01 *LBL1
02  STO4        t
03    R↓
04  STO3        P
05    R↓
06  STO2        sc/2 or sL or ss
07    R↓
08    4
09    ÷
10  STO1        D/4
11  X=0?
12  GTO9        Calculate D
13    R↓
14  X=0?
15  GTO8        Calculate t
16    R↓
17  X=0?
18  GTO7        Calculate P
19  RCL1        Calculate sc/2 or
20  RCL3          sL or ss
21    x
22  RCL4
23  GTO0
24 *LBL9
25  RCL4
26  RCL2
27    x
28    4
29    x
30  RCL3
31  GTO0
32 *LBL8
33  RCL1
34  RCL3
35    x
36  RCL2
37  GTO0
38 *LBL7
39  RCL4
40  RCL2
41    x
42  RCL1
43 *LBL0
44    ÷
45   R/S        *** D,Sx,P, or t
```

*** "Printx" may be inserted before "R/S".

REGISTERS					
0	1 D/4	2 S_x	3 P	4 t	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

STRESS IN THICK-WALLED CYLINDERS

This program calculates the radial and tangential components of normal stress for thick-walled, cylindrical, pressure vessels.



EQUATIONS:

$$s_r = \frac{r_i^2 P_i - r_o^2 P_o}{r_o^2 - r_i^2} - \frac{r_i^2 r_o^2 (P_i - P_o)}{r^2 (r_o^2 - r_i^2)}$$

$$s_t = \frac{r_i^2 P_i - r_o^2 P_o}{r_o^2 - r_i^2} + \frac{r_i^2 r_o^2 (P_i - P_o)}{r^2 (r_o^2 - r_i^2)}$$

where:

s_r is the radial component of stress;

s_t is the tangential component of stress;

r_i is the internal radius;

r_o is the outer radius;

r is the radius where calculated stresses occur;

P_i is the internal pressure;

P_o is the outside pressure.

EXAMPLE:

A cylinder has an inner radius of 1.00 inch and an outer radius of 2.00 inches. The inner pressure is 10,000 pounds per square inch and the outer pressure is 150 pounds per square inch. What are the values of radial and tangential stresses for radii of 1.00, 1.25, 1.75 and 2.00 inches?

SOLUTION:

1.00	GSB1	
2.00	ENT↑	
150.00	ENT↑	
1.00	ENT↑	
10000.00	R/S	
-10000.00	***	s_r psi
	X⇄Y	
16266.67	***	s_t psi
1.25	GSB1	
	R/S	
-5272.00	***	
	X⇄Y	s_r
11538.67	***	s_t
1.75	GSB1	
	R/S	
-1155.10	***	s_r
	X⇄Y	
7421.77	***	s_t
2.00	GSB1	
	R/S	
-150.00	***	s_r
	X⇄Y	
6416.67	***	s_t

5

REMARKS:

A negative stress indicates compression.

REFERENCE:

J.E. Shigley, Mechanical Engineering Design, McGraw Hill, 1963.

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter r	r	GSB	1	
3.	Enter the following:				
	Outside radius	r_o	ENT↑		
	Outside pressure	P_o	ENT↑		
	Inside radius	r_i	ENT↑		
	Inside pressure	P_i			
4.	Compute radial stress component		R/S		s_r
5.	Compute tangential stress component		x↔y		s_t
6.	For a different r, enter r and go to step 4	r	GSB	1	
7.	For a new case go to step 2				

Program Listings

01	*LBL1	
02	STO5	r
03	RCL4	
04	√X	
05	RCL3	
06	RCL2	
07	√X	
08	RCL1	
09	R/S	P_i r_i P_o r_o
10	X⇄Y	
11	X²	
12	STO2	
13	R↓	
14	STO1	
15	X⇄Y	
16	STO3	
17	-	P_i-P_o
18	R↓	
19	X²	
20	STO4	
21	x	$r_i^2 r_o^2$
22	X⇄Y	
23	R↓	
24	x	
25	RCL1	
26	RCL2	
27	x	
28	RCL3	
29	RCL4	
30	x	
31	-	
32	RCL4	
33	RCL2	
34	-	$r_o^2-r_i^2$
35	÷	
36	X⇄Y	
37	LSTX	$r_o^2-r_i^2$
38	RCL5	
39	X²	
40	x	
41	÷	
42	-	s_r
43	STO6	
44	LSTX	
45	2	
46	x	s_t
47	+	
48	RCL6	
49	R/S	s_r s_t

REGISTERS

0	1 P_i	2 r_i^2	3 P_o	4 r_o^2	5 r
6 s_r	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

MOHR CIRCLE FOR STRESS

Given the state of stress on an element, the principal stresses and their orientation can be found. The maximum shear stress and its orientation can also be found.



EQUATIONS:

$$s_{smax} = \sqrt{\left(\frac{s_x - s_y}{2}\right)^2 + s_{xy}^2}$$

6. $$s_1 = \frac{s_x + s_y}{2} + s_{smax}$$

$$s_2 = \frac{s_x + s_y}{2} - s_{smax}$$

$$\theta = 1/2 \tan^{-1}\left(\frac{2s_{xy}}{s_x - s_y}\right)$$

$$\theta_s = 1/2 \tan^{-1} - \left(\frac{s_x - s_y}{2s_{xy}}\right)$$

where:

s_{smax} is the maximum shear stress;

s_1 and s_2 are the principal normal stresses;

θ is the angle of rotation from the principal axis to the original axis;

θ_s is the angle of rotation from the axis of maximum shear stress to the original axis;

s_x is the stress in the x direction;

s_y is the stress in the y direction;

s_{xy} is the shear stress on the element.

REFERENCE:

Spotts, M.F., Design of Machine Elements, Prentic-Hall, 1971.

EXAMPLE:

If s_x = 25000 psi, s_y = -5000 psi, and s_{xy} = 4000 psi, compute the principal stresses and the maximum shear stress.



SOLUTION:

```
25000.00 ENT↑
-5000.00 ENT↑
 4000.00 GSB1
25524.17  ***
          R/S
-5524.17  ***
          R/S
    7.47  ***
          R/S
  -37.53  ***
          R/S
15524.17  ***
```

25524.17	***	s_1 (psi)
	R/S	
-5524.17	***	s_2 (psi)
	R/S	
7.47	***	θ (degrees)
	R/S	
-37.53	***	θ_s (degrees)
	R/S	
15524.17	***	s_{smax} (psi)

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter the following:				
	Stress in the x direction (negative for	S_x	ENT↑		
	compression)				
	Stress in the y direction (negative for	S_y	ENT↑		
	compression)				
	Shear stress	S_{xy}			
3.	Compute the following:				
	First principal stress		GSB	1	S_1
	Second principal stress		R/S		S_2
	Angle of rotation (principal)		R/S		θ
	Angle of rotation (shear)		R/S		θ_s
	Maximum shear stress		R/S		s_{smax}
	NOTE: Do not disturb the stack during				
	step 3				
4.	For a new case, go to step 2.				

Program Listings

*LBL1		
ENT↑		
R↓	s_{xy} s_y s_x s_{xy}	
STO3		
R↓		
X⇄Y		
STO1	s_x	
X⇄Y	s_y	
ST+1		
-	$s_x - s_y$	
2		
ST÷1		
÷		
STO4	$(s_x - s_y)/2$	
→P		
STO2	s_{smax}	
RCL1		
+		
R/S	** s_1	
X⇄Y	$2 \cdot \theta$	
RCL1		
RCL2		
-		
R/S	** s_2	
X⇄Y		
2		
÷		
R/S	** θ	
RCL4		
RCL3		
÷		
CHS		
TAN⁻¹		
2		
÷		
R/S	** θ_s	** "Printx" may replace "R/S".
RCL2		
R/S	*** s_{smax}	*** "Printx" may be inserted before "R/S".

REGISTERS

0	1 $(s_x+s_y)/2$	2 s_{smax}	3 s_{xy}	4 $(s_x-s_y)/2$	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

CIRCULAR PLATES WITH SIMPLY SUPPORTED EDGES

This program can be used to calculate the deflection and stress at the center of a simply supported circular plate with uniformly distributed or concentrated central loads.

EQUATIONS:

for a concentrated central load:

$$y_{max} = \frac{(3 + \mu)Pr^2}{16\pi(1 + \mu)D}$$

$$s_{max} = \frac{P}{h^2}\left[(1+\mu)\left(0.485 \ln \frac{r}{h} + 0.52\right) + 0.48\right]$$

for a uniformly distributed load:

$$y_{max} = \frac{(5 + \mu)Wr^4}{64D(1 + \mu)}$$

$$s_{max} = \frac{3(3 + \mu)Wr^2}{8h^2}$$

where:

7.

$$D = \frac{E h^3}{12(1 - \mu^2)}$$

y_{max} is the maximum deflection;

s_{max} is the maximum stress;

μ is Poisson's ratio;

E is the modulus of elasticity;

h is the thickness of the plate;

r is the radius of the plate;

W is the uniformly distributed load;

P is the concentrated central load.

REFERENCES:

Spotts, M.F., Design of Machine Elements, Prentice-Hall, Inc., 1971.

REMARKS:

Deflections must be small compared to thickness of plate.

EXAMPLE 1:

Assuming that a manhole cover with an automobile tire at its center may be modeled as a simply supported flat plate with concentrated central load, what is the deflection at the center of the plate? What is the stress?

$E = 30 \times 10^6$ psi

$h = 0.75$ in

$\mu = 0.3$

$r = 15$ in

$P = 1500$ lb

EXAMPLE 2:

A simply supported 1/4 inch thick plate ($E = 30 \times 10^6$, $\mu = 0.3$) withstands 50 pounds per square inch. If the radius is 5 inches, what is the deflection and what is the stress at the center of the plate?

SOLUTIONS:

```
(1)  30.+06 ENT↑          (2)  30.+06 ENT↑
       0.75 ENT↑                 0.25 ENT↑
       0.30 ENT↑                 0.30 ENT↑
      15.00 GSB1                 5.00 GSB1
    1500.00 GSB2                50.00 GSB3
       0.01  *** (in)            0.05  *** (in)
             R/S                       R/S
    8119.49  *** (psi)       24750.00  *** (psi)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Input modulus of elasticity	E	ENT↑		E
3.	Input thickness of plate	h	ENT↑		h
4.	Input Poisson's ratio	μ	ENT↑		
5.	Input radius of plate	r	GSB	1	
6.	If the load is distributed go to step 10				
7.	Input concentrated load and calculate				
	deflection	P	GSB	2	y_{max}
8.	Calculate maximum stress		R/S		s_{max}
9.	For new load go to step 7. For new case				
	go to step 2.				
10.	Input distributed load and calculate				
	deflection	W	GSB	3	y_{max}
11.	Calculate maximum stress		R/S		s_{max}
12.	For new load go to step 10. For new case				
	go to step 2.				

Program Listings

```
01 *LBL1
02   STO1
03     R↓
04   STO2
05     R↓
06   STO3
07      3
08     Yˣ
09      x
10      3
11      ÷
12      1
13   RCL2
14      -
15      ÷
16   STO4        4 D ( 1 + μ)
17    R/S
18 *LBL2
19   STO5        P
20   RCL1
21     X²
22      x
23      4
24      ÷
25     Pi
26      ÷
27      3
28 *LBL0
29   RCL2
30      +
31      x
32   RCL4
33      ÷
34    RTN        *** y_max
35   RCL1
36   RCL3
37      ÷
38     LN
39      .
40      4
41      8
42      5
43      x
44      .
45      5
46      2
47      +
48   RCL2
49      1
50      +
51      x
52      .
53      4
54      8
55      +
56   RCL5
57      x
58   RCL3
59     X²
60      ÷
61    R/S        *** s_max
62 *LBL3
63   STO6        W
64   RCL1
65      2
66      ÷
67      4
68     Yˣ        r⁴/16
69      x
70      5
71   GSB0
72    R/S
73   RCL1
74   RCL3
75      ÷
76     X²
77      8
78      ÷
79      3
80      x
81      3
82   RCL2
83      +
84      x
85   RCL6
86      x
87    R/S        *** s_max
```

*** "Printx" may be inserted before "RTN" or "R/S".

REGISTERS

0	1 r	2 μ	3 h	4 4 D(1+μ)	5 P
6 W	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

CIRCULAR PLATES WITH FIXED EDGES

This program can be used to calculate the maximum deflection and stress for a circular plate with fixed edges. Either central concentrated loads or distributed loads may be input.

EQUATIONS:

$$y_{max} = \frac{Pr^2}{16\pi D}$$

$$s_{max} = \frac{P}{h^2}(1+\mu)\left(0.485 \ln \frac{r}{h} + 0.52\right)$$

for distributed loads:

$$y_{max} = \frac{Wr^4}{64D}$$

$$s_{max} = \frac{3Wr^2}{4h^2} \quad \text{(at edge of plate)}$$

where:

$$D = \frac{Eh^3}{12(1-\mu^2)}$$

y_{max} is the maximum deflection

s_{max} is the maximum stress;

P is the concentrated load;

W is the distributed load;

r is the radius of the plate;

h is the thickness of the plate;

μ is Poisson's ratio;

E is the modulus of elasticity.

REFERENCE:

Spotts, M.F., Design of Machine Elements, Prentice-Hall, Inc., 1971.

REMARKS:

Deflections must be small compared to the thickness of plate.

EXAMPLE 1:

The cap on a pressure vessel is a 1/4 inch thick steel plate ($E = 30 \times 10^6$ psi, $\mu = 0.3$) with a 6 inch radius. It is clamped to the opening of the pressure vessel by a ring of bolts. What are the maximum and minimum deflections and stresses in the plate if pressure cycles from 50 to 60 psi?

EXAMPLE 2:

An adjustable focal length mirror is to derive its concaved shape due to a variable force applied at its center. The mirror is chrome plated steel ($E = 30 \times 10^6$ psi, $\mu = 0.3$), 0.1 inches thick and has a radius of 12 inches. What is the deflection of the center for a force of 6.0 pounds. The edges are held securely.

SOLUTIONS:

(1)

```
   30.+06 ENT↑
     0.25 ENT↑
     0.30 ENT↑
     6.00 GSB1
    50.00 GSB3
     0.02  ***  (in)min
           R/S
 21600.00  ***  (psi)
    60.00 GSB3
     0.03  ***  (in)max
           R/S
 25920.00  ***  (psi)
```

(2)

```
   30.+06 ENT↑
     0.10 ENT↑
     0.30 ENT↑
    12.00 GSB1
     6.00 GSB2
          FIX5
  0.00626  ***  (in)
```

8.

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Input modulus of elasticity	E	ENT↑		E
3.	Input thickness of plate	h	ENT↑		h
4.	Input Poisson's ratio	μ	ENT↑		μ
5.	Input radius of plate	r	GSB	1	D
6.	If the load is distributed go to step 10				
7.	Input concentrated load and calculate				
	deflection	P	GSB	2	y_{max}
8.	Calculate maximum stress		R/S		s_{max}
9.	For new load go to step 7. For new case				
	go to step 2				
10.	Input distributed load and calculate				
	deflection	W	GSB	3	y_{max}
11.	Calculate maximum stress		R/S		s_{max}
12.	For new load go to step 10. For new case				
	go to step 2.				

Program Listings

Step	Key	Comment
01	*LBL1	
02	STO1	
03	R↓	
04	STO2	
05	R↓	
06	STO3	
07	3	
08	Y^X	
09	×	
10	1	
11	2	
12	÷	
13	1	
14	RCL2	
15	X^2	
16	−	
17	÷	
18	STO4	D
19	R/S	
20	*LBL2	P
21	STO5	
22	RCL1	
23	4	
24	÷	
25	X^2	$r^2/16$
26	×	
27	RCL4	
28	÷	
29	Pi	
30	÷	
31	R/S	*** y_{max}
32	RCL1	
33	RCL3	
34	÷	
35	LN	
36	.	
37	4	
38	8	
39	5	
40	×	
41	.	
42	5	
43	2	
44	+	
45	1	
46	RCL2	
47	+	
48	×	
49	RCL5	
50	×	
51	RCL3	
52	X^2	
53	÷	
54	R/S	*** s_{max}
55	*LBL3	W
56	STO6	
57	RCL1	
58	2	
59	÷	
60	4	
61	Y^X	$r^4/16$
62	×	
63	4	
64	÷	
65	RCL4	
66	÷	
67	R/S	** y_{max}
68	RCL1	
69	2	
70	÷	
71	RCL3	
72	÷	
73	X^2	$r^2/4h^2$
74	3	
75	×	
76	RCL6	
77	×	
78	R/S	*** s_{max}

** "Printx" may be replace "R/S".

*** "Printx" may be inserted before "R/S".

REGISTERS

0	1 r	2 μ	3 h	4 D	5 P
6 W	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

COMPRESSIVE BUCKLING

This program performs an interchangeable solution for the four properties of slender compression members or columns: P_{cr}, the critical buckling load; E, the modulus of elasticity; I, the minimum moment of inertia; and ℓ, the length of the member.

EQUATIONS:

Three configurations are possible, identified by the number of fixed ends on the member: 0, both ends hinged; 1, one end free and one fixed; 2, both ends fixed.



$$P_{cr} = \frac{\pi^2 EI}{\ell^2}$$

$$P_{cr} = \frac{\pi^2 EI}{4\ell^2}$$

$$P_{cr} = \frac{4\pi^2 EI}{\ell^2}$$

REMARKS:

Uncertainties such as the amount of restraint at the ends, eccentricity of the load, initial warp, nonhomogeneity of the material and deflection caused by lateral loads, can cause very significant changes in the behavior of a compressive member.

EXAMPLE 1:

If an 8 inch steel (E = 30 x 10^6 psi) piston rod (a piston rod has zero fixed ends) must withstand a load of 15000 pounds without buckling, what moment of inertia must it have?

EXAMPLE 2:

Steel columns 40 feet long are used to support a bridge. What is the maximum load that the column can withstand without buckling? Assume 1 fixed end. E = 30 x 10^6 psi, I = 700 in^4.

SOLUTIONS:

(1)
```
     0.00 GSB1
 15000.00 ENT↑
   30.+06 ENT↑
     0.00 ENT↑
     8.00 GSB2
  3.24-03  ***  I
```

(2)
```
     1.00 GSB1
     0.00 ENT↑
   30.+06 ENT↑
   700.00 ENT↑
   480.00 GSB2
224893.33  ***  P
```

9

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Select column geometry by entering the	0,1, or 2	GSB	1	
	number of fixed ends				
3.	Enter the following (enter a zero for the				
	unknown quantity)				
	Vertical load	P	ENT↑		
	Modulus of elasticity	E	ENT↑		
	Moment of inertia	I	ENT↑		
	Length of column	ℓ			
4.	Compute the unknown quantity		GSB	2	P,E,I,orℓ^2
5.	For another calculation, go to step 2				

Program Listings

```
01 *LBL1        0,1, or 2
02     5
03  STO0        i = 5
04    R↓
05    Pi
06    X²
07  STO1
08    R↓
09  X=0?
10   R/S        c = 1
11     2
12  X≠Y?
13   1/X
14    X²        c = 1/4 or 4
15  ST×1
16   R/S
17 *LBL2        ℓ I E P
18    X²
19  STO5
20    R↓
21  STO2
22    R↓
23  STO3
24    R↓
25  STO4
26  X=0?
27  GTO9        Calculate P
28   DSZ
29    R↓        i = 4
30  X=0?
31  GTO9        Calculate ℓ²
32    R↓
33   DSZ        i = 3
34  X≠0?
35   DSZ        Calculate E,i=2
36  RCL4        Calculate I,i=3
37  RCL5
38     ×
39  RCL1
40     ÷
41  GTO0
42 *LBL9
43  RCL1
44  RCL2
45  RCL3
46     ×
47     ×
48 *LBL0
49  RCLi
50     ÷
51   R/S        *** ℓ²,E,I, or P
```

*** R/S may be inserted before "R/S".

REGISTERS					
0 i	1 $c\pi^2$	2 I	3 E	4 P	5 ℓ^2
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

ECCENTRICALLY LOADED COLUMNS

This program calculates the maximum deflection, the maximum moment, and the maximum stress in an eccentrically loaded column under compressive stress.



EQUATIONS:

$$y_{max} = e\left[\sec \frac{\ell}{2}\sqrt{\frac{P}{EI}} - 1\right]$$

$$M_{max} = P\,[e + y_{max}]$$

$$s_{max} = \frac{P}{A}\left[1 + \frac{ecA}{I}\sec \frac{\ell}{2}\sqrt{\frac{P}{EI}}\right]$$

where:

- y_{max} is the maximum deflection;
- e is the eccentricity;
- ℓ is the column length;
- P is the compressive load;
- E is the modulus of elasticity;
- I is the moment of inertia;
- M_{max} is the maximum internal moment;
- s_{max} is the maximum normal stress in the column;
- c is the distance from the neutral axis of the column to the outer surface;
- A is the area of the cross section

REMARKS:

Columns must be of constant cross section. Stresses may not exceed the elastic limit of the material.

REFERENCE:

Spotts, M.F., Design of Machine Elements, Prentice-Hall, 1971.

EXAMPLE:

A column 50 feet long is to support 8000 pounds. The load is to be offset 6 inches. What are the maximum values of deflection, moment, and stress in the member?

$E = 30 \times 10^6$

$I = 107 \text{ in}^4$

$A = 7 \text{ in}^2$

$c = 2 \text{ in}$

SOLUTION:

```
               RAD
     107.00  ST01
     30.+06  ST02
      50.00  ENT↑
      12.00     x
             ST03
       6.00  ST04
    8000.00  ST05
             GSB1
       0.74   ***   (in)
             GSB2
   53936.76   ***   (in-lb)
       2.00  ENT↑
       7.00  GSB3
    2151.02   ***   (psi)
```

10

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Initialize		g	RAD	
3.	Store data:				
	Moment of inertia	I	STO	1	
	Modulus of elasticity	E	STO	2	
	Length of column	ℓ	STO	3	
	Eccentricity	e	STO	4	
	Load	P	STO	5	
4.	To calculate maximum deflection		GSB	1	y_{max}
5.	To calculate maximum moment		GSB	2	M_{max}
6.	To calculate maximum stress:				
6a.	Enter distance from neutral axis	c	ENT↑		
6b.	Enter section area and run	A	GSB	3	s_{max}
7.	For a new case, go to step 3 and store different value(s).				

Program Listings


```
01 *LBL1
02  GSB0
03   R/S        *** y_max
04 *LBL2
05  GSB0
06  RCL4
07    +
08  RCL5
09    x
10   R/S        *** M_max
11 *LBL0
12  RCL5
13  RCL2
14    ÷
15  RCL1
16    ÷
17   √X
18  RCL3
19    x
20    2
21    ÷
22   COS
23   1/X        sec (x)
24    1
25    -
26  RCL4
27    x
28   RTN
29 *LBL3
30  STO7
31    x         cA
32  RCL4
33    x
34  RCL1
35    ÷
36  ENT↑
37  RCL5
38  RCL2
39    ÷
40    4
41    ÷
42  RCL1
43    ÷
44   √X
45  RCL3
46    x
47   COS
48   1/X
49    x
50    1
51    +
52  RCL5
53  RCL7
54    ÷
55    x
56   R/S        *** s_max
```

*** "Printx" may be inserted before "R/S".

REGISTERS

0	1 I	2 E	3 ℓ	4 e	5 P
6	7 A	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

In the Hewlett-Packard tradition of supporting HP programmable calculators with quality software, the following titles have been carefully selected to offer useful solutions to many of the most often encountered problems in your field of interest. These ready-made programs are provided with convenient instructions that will allow flexibility of use and efficient operation. We hope that these Solutions books will save your valuable time. They provide you with a tool that will multiply the power of your HP-19C or HP-29C many times over in the months or years ahead.

Mathematics Solutions
Statistics Solutions
Financial Solutions
Electrical Engineering Solutions
Surveying Solutions
Games
Navigational Solutions
Civil Engineering Solutions
Mechanical Engineering Solutions
Student Engineering Solutions

HEWLETT hp PACKARD

Reorder No. 00029-14008 Printed in U.S.A. 00029-90012